
 Premiers pas en Maple (Partie 4 de 5)
© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

Pour voir le contenu des différentes sections, cliquer avec la souris sur le triangle  précédant le titre. La section 
se déployera  et son contenu sera affiché.
 (un clic gauche ou un clic droit sur le triangle permettra de rétracter la section et son contenu sera alors masqué).

Pour votre confort, vous pouvez ajuster la taille de l'affichage à l'aide de la commande Facteur de zoom du menu
Affichage.

Bonne lecture à tous !

 Ce document Maple est exécutable avec la version 2020.2

Initialisation
restart;

with(plots,display,implicitplot3d,spacecurve);

Objectif
Cette quatrième partie a comme objectif d'initier l'élève aux tracés de courbes et de plans dans l'espace. Le 
tracé de courbes dans l'espace se limitera aux tracés de l'intersections de plans parallèles aux plans de 
coordonnées et de quadriques.

Syntaxe paramétrique de tracés de surfaces dans l'espace
Les options permises pour le tracé d'une surface dans l'espace sont trop nombreuses pour être exposées dans 
ce document. Le lecteur les retrouverons dans l'aide plot3d,options. Tout de même, les nombreux exemples de
ce document donneront un bon aperçu de leur utilité. La macro-commande setoptions et setoptions3d de 
l'extension plots, permet de configurer une séquence d'options global pour les tracés 2D et 3D mais deans ce
document d'initiation, cette  macro-commande ne sera pas utilisée.

 Surfaces définies explicitement
Tracé d'un plan parallèle à deux axes de coordonnées
Il suffit de préciser une coordonnée constante et de faire varier les deux autres.

Plan d'équation . Fixons l'abscisse et faisons varier l'ordonnée et la cote.
P1:=plot3d([2,y,z],z=-5..5,y=-5..5,

      axes=normal,

      style=patchnogrid,

      color="Dalton Vermillion",

      lightmodel=none):

P1;

Plan d'équation . Fixons l'ordonnée et faisons varier l'abscisse et la cote.
P2:=plot3d([x,3,z],x=-8..8,z=-6..6,

      axes=normal,

      style=patchnogrid,

      color="Dalton Yellow",

      lightmodel=none):

P2;

Plan d'équation . Fixons la cote et faisons varier l'ordonnée et la cote.
P3:=plot3d([x,y,2],x=-5..8,y=-5..5,

      axes=normal,

      style=patchnogrid,

      color="Dalton Blue",

      lightmodel=none):

P3;

Superposons maintenant ces trois plans dans un même graphique.
display([P1,P2,P3]);

Tracé d'un plan parallèle à un axe de coordonnées
Il suffit de préciser une coordonnée dépendante d'une (seule) autre et faire varier ensuite les deux 
coordonnées indépendamment.

Plan d'équation .
plot3d([x,-5*x+3,z],x=-0.5..1,z=-4..4,

  axes=boxed,lightmodel=none,

  orientation=[15,60],

  style=patchnogrid,

  color="Dalton BluishGreen");

Tracé d'un plan parallèle à aucun axe de coordonnées
Soit le plan d'équation .

Éq:=x+y+z/3-1=0;

isolate(Éq,z);

Plan:=plot3d([x,y,-3*x-3*y+3],x=-1..1,y=-1 ..2,

        axes=normal,lightmodel=none,

        orientation=[70,70],

        style=surface,

        color="Niagara Burgundy"):

        #transparency=0.05,

        #view=[-1..2,-1..0.75,0..4]

Plan;

> > Plan_xOy:=plot3d([x,y,0],x=0..2,y=0 ..3,

      style=patchnogrid,transparency=0.25,

      color="Dalton BluishGreen"):

Plan_xOz:=plot3d([x,0,z],x=0..2,z=0 ..6,

      style=patchnogrid,transparency=0.25,

      color="Dalton BluishGreen"):

Plan_yOz:=plot3d([0,y,z],y=0..4,z=0 ..6,

      style=patchnogrid,transparency=0.25,

      color="Dalton BluishGreen"):

display([Plan,Plan_xOy,Plan_xOz,Plan_yOz]);

Visualisons ce plan dans le premier octant.
Plan_bis:=plot3d([x,y,-3*x-3*y+3],x=0..1.2,y=0 ..1.2,

      style=patchnogrid,

      color="Niagara Burgundy"):

display(Plan_bis,view=[0..1.2,0..1.2,0..3.5],axes=normal);

Voici comment obtenir la trace de ce plan dans le plan de coordonnés .
Trace_xOz:=spacecurve([x,0,-3*x+3],x=0..1,thickness=3,color=

yellow):

display(Trace_xOz,axes=normal);

Faisons de même pour les traces dans les deux autres plans de coordonnées et superposons les trois 
traces avec le plan lui-même.

Trace_xOy:=spacecurve([x,-x+1,0],x=0..1,thickness=3,color=

yellow):

Trace_yOz:=spacecurve([0,y,-3*y+3],y=0..1,thickness=3,color=

yellow):

display(Trace_xOz,Trace_xOy,Trace_yOz,axes=normal);

display([Plan,Plan_xOy,Plan_xOz,Plan_yOz,Trace_xOz,Trace_xOy,

Trace_yOz],axes=normal);

Tracé d'une surface cylindrique
Traçons la surface d'équation .

f:=(x,y)->y^2;

Surface:=plot3d([x,y,f(x,y)],x=-5..5,y=-5..5,

         style=patchnogrid,

         axes=framed,

         orientation=[40,70]):

Surface;

Plan1:=plot3d([2,y,z],y=-5..5,z=-5..30,style=patchnogrid):

Trace1:=spacecurve([2,y,f(2,y)],y=-5..5,color=navy,thickness=2):

Plan2:=plot3d([-3,y,z],y=-5..5,z=-5..30,style=patchnogrid):

Trace2:=spacecurve([-3,y,f(-3,y)],y=-5..5,color=navy,thickness=

2):

Surface:=plot3d([x,y,f(x,y)],x=-5..5,y=-5..5,style=patchnogrid):

display(Plan1,Plan2,Trace1,Trace2,Surface,axes=framed,

orientation=[40,70]);

f:=(x,y)->sin(x);

Surface:=plot3d([x,y,f(x,y)],x=-2*Pi..2*Pi,y=-1..1,style=

surface,grid=[80,60]):

display(Surface,axes=framed,lightmodel=light3,color="Dalton 

BluishGreen");

Surface:=plot3d([x,y,f(x,y)],x=-2*Pi..2*Pi,y=-1..1,style=

surface,grid=[80,60],transparency=0.05):

Plan1:=plot3d([x,-0.5,z],x=-2*Pi..2*Pi,z=-1.5..1.5,style=

surface,lightmodel=light3,transparency=0.15):

Trace1:=spacecurve([x,-0.5,f(x,y)],x=-2*Pi..2*Pi,thickness=2,

color=yellow):

display([Surface,Plan1,Trace1],axes=framed,orientation=[72,49],

color="Dalton BluishGreen");

Tracé d'un paraboloïde elliptique

Traçons le paraboloïde d'équation  .

f:=(x,y)->20+x^2/4-y^2/5;

Lieu:=plot3d([x,y,f(x,y)],x=-5..5,y=-sqrt(25-x^2)..sqrt(25-x^2),

           color="Bright Cyan",

           grid=[160,120]):

Domaine:=plot3d([x,y,f(0,5)],x=-5..5,y=-sqrt(25-x^2)..sqrt(25-

x^2),

         grid=[80,20],

         style=patchnogrid,

         color=grey):

display(Lieu,Domaine,

      axes=normal,

      style=surfacecontour,contours = 25,
      orientation=[60,50,0]);

Plan1:=plot3d([x,y,22],x=-4..4,y=-4..4,style=surface,color=

"Bright Cyan",transparency=0.25):

display([Lieu,Domaine,Plan1],

         axes=framed,

         orientation=[60,75,0]);

Trace1:=spacecurve([x,1/2*sqrt(-40+5*x^2),22,x=sqrt(8)+10E-6..

sqrt(8)+1.1],thickness= 4,color=yellow):

Trace2:=spacecurve([x,-1/2*sqrt(-40+5*x^2),22,x=sqrt(8)+10E-6..

sqrt(8)+1.1],thickness=4,color=yellow):

Trace3:=spacecurve([x,1/2*sqrt(-40+5*x^2),22,x=-sqrt(8)-10E-6..-

sqrt(8)-1.1],thickness= 4,color=yellow):

Trace4:=spacecurve([x,-1/2*sqrt(-40+5*x^2),22,x=-sqrt(8)-10E-6..

-sqrt(8)-1.1],thickness=4,color=yellow):

display([Domaine,Lieu,Plan1,Trace||(1..4)],

         axes=framed,

         orientation=[60,65,0]);

Plan2:=plot3d([2,y,z],y=-6..6,z=15..26,style=surface,color=

"Bright Cyan",transparency=0.15):

display([Lieu,Domaine,Plan2],

         axes=framed,

         orientation=[65,80,0]);

Trace_yOz:=spacecurve([2,y,21-1/5*y^2,y=-sqrt(25-2^2)..sqrt(25

-2^2)],

           thickness=4,color=yellow):

display([Lieu,Domaine,Plan2,Trace_yOz],

         axes=framed,

         orientation=[60,63]);

Plan3:=plot3d([x,3,z],x=-6..6,z=15..26,style=surface,color=

"Bright Cyan",transparency=0.15):

display([Lieu,Domaine,Plan3],

         axes=framed,

         orientation=[50,75]);

Trace_xOz:=spacecurve([x,3,91/5+1/4*x^2,x=-4..4],thickness=4,

color=yellow):

display([Lieu,Domaine,Plan3,Trace_xOz],

         axes=framed,

         orientation=[60,65,0]);

 Surfaces définies implicitement
Tracé d'un hyperbolïde à une nappe
Traçons la surface hyperbolique à deux nappes d'équation .

Lieu:=implicitplot3d(x^2+y^2-z^2=4,x=-5..5,y=-5..5,z=-5..5,grid=

[40,40,40],

       view=[-6..6,-6..6,-4..4],style=wireframe,axes=framed,

color="Dalton BluishGreen"):

Lieu;

Plan1:=plot3d([2.5,y,z],y=-4..4,z=-4..4,style=patchnogrid,color=

"Bright Purple"):

Plan2:=plot3d([x,3,z],x=-4..4,z=-4..4,style=patchnogrid,color=

"Bright Orange"):

display([Lieu,Plan1,Plan2],orientation=[25,60,0],lightmodel=

none);

Tracé  d'un hyperbolïde à deux nappes

Traçons la surface hyperbolique à deux nappes d'équation .

Éq_Lieu:=x^2/9-y^2/16-z^2/25=1;

Lieu:=implicitplot3d(Éq_Lieu,x=-10..10,y=-15..15,z=-20..20,

      grid=[40,40,40],

      style=patchnogrid,

      axes=framed):

display(Lieu,orientation=[60,50],color="Dalton BluishGreen");

Plan1:=plot3d([x,y,5],x=-10..10,y=-15..15,style=patchnogrid,

color="Nautical PaleGreen",transparency=0.10):

display([Lieu,Plan1],orientation=[58,50],color="Dalton 

BluishGreen");

Plan2:=plot3d([x,5,z],x=-10..10,z=-15..15,style=patchnogrid,

color="Nautical PaleGreen",transparency=0.20):

display([Lieu,Plan2],orientation=[55,50,0],color="Dalton 

BluishGreen");

Trace1:=spacecurve([t,4/3*sqrt(t^2-18),5],t=sqrt(18)+10E-6..sqrt

(18)+5.8,color=yellow,thickness=2):

Trace2:=spacecurve([t,-4/3*sqrt(t^2-18),5],t=sqrt(18)+10E-6..

sqrt(18)+5.8,color=yellow,thickness=2):

Trace3:=spacecurve([t,4/3*sqrt(t^2-18),5],t=-sqrt(18)-5.8..-sqrt

(18)-10E-6,color=yellow,thickness=2):

Trace4:=spacecurve([t,-4/3*sqrt(t^2-18),5],t=-sqrt(18)-5.8..-

sqrt(18)-10E-6,color=yellow,thickness=2):

display([Lieu,Plan1,Trace||(1..4)],orientation=[55,50],color=

"Dalton BluishGreen",transparency=0.10);

Trace1:=spacecurve([t,5,5/12*sqrt(16*t^2-369)],t=sqrt(369/16)

+10E-6..sqrt(369/16)+5,color=yellow,thickness=2):

Trace2:=spacecurve([t,5,-5/12*sqrt(16*t^2-369)],t=sqrt(369/16)

+10E-6..sqrt(369/16)+5,color=yellow,thickness=2):

Trace3:=spacecurve([t,5,5/12*sqrt(16*t^2-369)],t=-sqrt(369/16)

-5..-sqrt(369/16)-10E-6,color=yellow,thickness=2):

Trace4:=spacecurve([t,5,-5/12*sqrt(16*t^2-369)],t=-sqrt(369/16)

-5..-sqrt(369/16)-10E-6,color=yellow,thickness=2):

display([Lieu,Plan2,Trace||(1..4)],orientation=[55,50,0],color=

"Dalton BluishGreen",transparency=0.10);

Tracé d'un ellipsoïde

Soit l'ellipsoïde d'équation   .

Ellipsoïde:=implicitplot3d(x^2/8+1/25*y^2+1/9*z^2-1 = 0,x=-4..4,

y=-5..5,z=-3..3,

      style=patchnogrid,grid=[30,60,30]):

display(Ellipsoïde,axes=framed,color="Bright Cyan",glossiness=

0.3,scaling=constrained,orientation=[55,75,0]);

Obtenons maintenant les traces de trois plans parallèles aux plans de coordonnées sur cet ellipsoïde.

Plan_xOz:=plot3d([x,3,z],x=-4..4,z=-4..4,style=patchnogrid,

color="Dalton ReddishPurple",transparency=0.10):

display(Ellipsoïde,Plan_xOz,axes=framed,color="Bright Cyan",

glossiness=0.3,scaling=constrained);

Trace1:=spacecurve([sqrt(8*16/25)*cos(t),3,sqrt(9*16/25)*sin(t)

],t=0..2*Pi,color=yellow,thickness=3):

display(Ellipsoïde,Trace1,axes=framed,color="Bright Cyan",

glossiness=0.3,scaling=constrained);

> > Plan_yOz:=plot3d([2,y,z],y=-5..5,z=-4..4,style=patchnogrid,

color="Dalton ReddishPurple",transparency=0.10):

display(Ellipsoïde,Plan_yOz,axes=framed,color="Bright Cyan",

glossiness=0.3,scaling=constrained);

Trace2:=spacecurve([2,sqrt(25/2)*cos(t),sqrt(9/2)*sin(t)],t=0.

.2*Pi,color=yellow,thickness=3):

display(Ellipsoïde,Trace2,axes=framed,color="Bright Cyan",

glossiness=0.3,scaling=constrained);

> > Plan_xOy:=plot3d([x,y,2],x=-5..5,y=-5..5,style=patchnogrid,

color="Dalton ReddishPurple",transparency=0.10):

display(Ellipsoïde,Plan_xOy,axes=framed,color="Bright Cyan",

glossiness=0.3,scaling=constrained);

Trace3:=spacecurve([sqrt(8*5/9)*cos(t),sqrt(25*5/9)*sin(t),2],t=

0..2*Pi,color=yellow,thickness=3):

display(Ellipsoïde,Trace3,axes=framed,color="Bright Cyan",

glossiness=0.3,scaling=constrained);

Tracé d'une sphère

Sphère:=implicitplot3d(x^2+y^2+z^2 = 25,x=-5..5,y=-5..5,z=-5..5,

style=patchnogrid,grid=[25,25,25],

axes=normal,scaling=constrained,axes=framed,color="Bright Cyan",

glossiness=0.3,orientation=[50,75,0]);

 Illustrons la trace sur la sphère du plan .
Trace1:=spacecurve([3*cos(t),4,3*sin(t)],t=0..2*Pi,color=yellow,

thickness=3):

display(Sphère, Trace1);

Tracé d'un cône elliptique
Cône:=implicitplot3d(x^2+y^2-z^2 = 0,x=-4..4,y=-4..4,z=-4..4,

style=patchnogrid,grid=[30,30,30],

orientation=[40,70],color="Bright Cyan",glossiness=0.3);

Trace1:=spacecurve([2*tan(t),2,-2/cos(t)],t=-Pi/3..Pi/3,color=

yellow,thickness=3):

Trace2:=spacecurve([2*tan(t),2,2/cos(t)],t=-Pi/3..Pi/3,color=

yellow,thickness=3):

display(Cône, Trace1,Trace2,transparency=0.1,axes=normal,

scaling=constrained);

